

WebFOCUS HTML Report Accessibility Support

In this section:

Controlling HTML Code For Accessibility
Centering a Heading
Summarizing Report or Graph Information
Describing Drill-Down Information
Adding Descriptive Text to an Image
On Demand Paging (ODP) WebViewer Support
Accessibility Limitations

WebFOCUS HTML report accessibility provides support for assistive technologies, such as screen readers.

WebFOCUS HTML report output complies with accessibility requirements as a result of the following features:

- ❑ A SET command that activates accessibility changes to WebFOCUS HTML output code.
- ❑ A SUMMARY attribute in a WebFOCUS StyleSheet that describes information contained in a report or graph.
- ❑ An ALT StyleSheet attribute in a drill-down report or graph that describes the detail information you can select.
- ❑ An ALT StyleSheet attribute that describes an image embedded in a report.
- ❑ A scope attribute included in the <TH> tag for column titles, starting in Release 7 Version 6.9.

Using the up and down arrow keys to navigate HTML report output might produce undesirable results, depending upon the content of the report. WebFOCUS HTML report accessibility provides support for assistive technologies, such as screen readers, when the report request complies with the accessibility requirements. In order to read 508 enabled reports effectively with Jaws®, it is recommended to use the table reading commands. Refer to the Freedom Scientific website (<http://www.freedomscientific.com/>) for specific commands for your version of Jaws.

For browser compatibility, refer to the Information Builders Web Browser Support for WebFOCUS documentation here:

https://techsupport.informationbuilders.com/tech/wbf/wbf_tmo_027.html

Information Builders accessibility support for Jaws is version 12.0 or higher. See the Freedom Scientific website (<http://www.freedomscientific.com/>) for browser requirement information.

Controlling HTML Code For Accessibility

How to:

Control HTML Code Accessibility

Example:

Controlling HTML Code for Accessibility

The SET ACCESSHTML command controls the generation of HTML code for page headings, page footings, sort headings, sort footings, and column titles in a WebFOCUS report. It also controls the HTML code generated for strong emphasis (boldface) and italics.

When SET ACCESSHTML is set to 508, the generated code supports products for disabled accessibility.

Syntax: **How to Control HTML Code Accessibility**

- ❑ For all requests in a procedure or for profile implementation:

```
SET ACCESSHTML = {508|OFF}
```

- ❑ For one request:

```
ON TABLE SET ACCESSHTML {508|OFF}
```

or

```
ON GRAPH SET ACCESSHTML {508|OFF}
```

where:

508

Generates HTML code compliant with Section 508 accessibility requirements.

This setting automatically turns off pagination, so a report appears as a single page. Column titles and headings are displayed only once at the top, and footings are displayed only once at the bottom.

OFF

Generates HTML code that is non-compliant with Section 508 accessibility requirements. This value is the default.

Note: Prior to Version 7 Release 7.04, the SET ACCESSIBLE command was the only accessibility command supported. Starting with Version 7 Release 7.04, the SET ACCESSHTML command became available.

For upward compatibility with releases prior to Version 7 Release 7.04, the following legacy commands are still available:

```
SET ACCESSIBLE = 508
ON TABLE SET ACCESSIBLE 508
```

Example: Controlling HTML Code for Accessibility

The SET command that enables accessibility appears in boldface:

```
TABLE FILE EMPLOYEE
HEADING
"SALARY REPORT"
"FOR COMPANY EMPLOYEES"
FOOTING
"FOR INTERNAL USE ONLY"
PRINT EMP_ID
BY DEPARTMENT
BY SALARY
ON DEPARTMENT SUBFOOT
"<DEPARTMENT SALARY TOTAL: <ST.SALARY "
" "
ON TABLE SET ACCESSHTML 508
ON TABLE SET STYLE *
TYPE=HEADING, LINE=1, STYLE=BOLD, $
TYPE=HEADING, LINE=2, STYLE=ITALIC, $
TYPE=REPORT, GRID=OFF, $
END
```

WebFOCUS generates the following HTML source code when SET ACCESSHTML is set to 508.

Page Heading

A page heading in a WebFOCUS report appears in a separate HTML table outside the main report table:

```
<TABLE CELLPADDING=0 WIDTH="100%"><TR>
<TD NOWRAP COLSPAN=3>
&nbsp;   </TD>
</TR>
<TR>
<TD NOWRAP COLSPAN=3>
<STRONG>SALARY REPORT</STRONG></TD>
</TR>
<TR>
<TD NOWRAP COLSPAN=3>
<EM>FOR COMPANY EMPLOYEES</EM></TD>
</TR>
</TABLE>
```

```
<TABLE CELLPADDING=1>
.
. (main report table)
.
</TABLE>
```

Text Formatting

The HTML tag is used instead of the tag to designate boldface text and the HTML tag is used instead of the <I> tag to designate italic text in the WebFOCUS StyleSheet:

```
<STRONG>SALARY REPORT</STRONG></TD>
</TR>
<TR>
<TD NOWRAP COLSPAN=3>
<EM>FOR COMPANY EMPLOYEES</EM></TD>
</TR>
```

Page Footing

A page footing in a WebFOCUS report appears in a separate HTML table outside the main report table:

```
<TABLE CELLPADDING=1>
.
. (main report table)
.
</TABLE>

<TABLE BORDER=0 CELLSPACING=0 CELLPADDING=0>
<TR>
<TD NOWRAP COLSPAN=3>
FOR INTERNAL USE ONLY</TD>
</TR>
</TABLE>
```

Sort Heading/Footing

A sort heading/footing in a WebFOCUS report appears in an HTML table embedded in the main report table:

```
<TABLE CELLPADDING=1>
.
. (main report table)
.
<TABLE BORDER=0 CELLSPACING=0 CELLPADDING=0 WIDTH="100%">
<TR>
<TD NOWRAP>
MIS SALARY TOTAL: $160,177.00 </TD>
</TR>
<TR>
<TD NOWRAP>
 </TD>
</TR>
</TABLE></TD></TR>
.
. (main report table)
.
>
```

Column Title

A column title in a WebFOCUS report appears in the HTML <TH> tag. In Version 7 Release 6.9, the scope attribute was added to the <TH> tag:

```
<TR>
<TH scope='col' NOWRAP ALIGN=LEFT VALIGN=BOTTOM>
<U>DEPARTMENT</U></TH>
<TH scope='col' NOWRAP ALIGN=RIGHT VALIGN=BOTTOM>
<U>SALARY</U></TH>
<TH scope='col' NOWRAP ALIGN=LEFT VALIGN=BOTTOM>
<U>EMP_ID</U></TH>
</TR>
```

The report is:

PAGE 1

SALARY REPORT
FOR COMPANY EMPLOYEES

<u>DEPARTMENT</u>	<u>SALARY</u>	<u>EMP_ID</u>
MIS	\$8,650.00	543729165
	\$9,000.00	543729165
	\$13,200.00	112847612
	\$17,750.00	117593129
	\$18,480.00	117593129
		219984371
	\$21,780.00	326179357
	\$25,775.00	818692173
	\$27,062.00	818692173

MIS SALARY TOTAL: \$160,177.00

PRODUCTION	\$9,050.00	119265415
	\$9,500.00	119265415
	\$10,000.00	071382660
	\$11,000.00	071382660
	\$15,000.00	451123478
	\$16,100.00	451123478
	\$21,120.00	126724188
	\$24,420.00	123764317
	\$26,862.00	123764317
	\$29,700.00	119329144

PRODUCTION SALARY TOTAL: \$172,752.00

FOR INTERNAL USE ONLY

Note:

- The column titles do not display in boldface by default.
- The JUSTIFY option for column titles is supported.

Centering a Heading

How to:

Center a Heading

Example:

Centering a Heading

The heading is located outside of the main table in a WebFOCUS HTML Section 508 accessible report. The column numbers and widths in the heading table and main table were synchronized to support centering of the heading above the columns in the main table. HEADING CENTER centers the heading over the body of the report by adding the COLSPAN and ALIGN=CENTER attributes to the <TD> tag for the heading rows in the heading table.

Syntax: How to Center a Heading

```
HEADING CENTER  
"headingline1"  
"headingline2"  
.  
.  
.
```

where:

headingline1, headingline2

Are lines of heading text enclosed in double quotation marks.

Example: Centering a Heading

This request uses the HEADING CENTER command to center the two heading lines:

```
TABLE FILE EMPLOYEE
HEADING CENTER
" SALARY REPORT "
" FOR COMPANY EMPLOYEES "
FOOTING
" FOR INTERNAL USE ONLY "
PRINT EMP_ID
BY DEPARTMENT
BY SALARY
ON DEPARTMENT SUBFOOT
"<DEPARTMENT SALARY TOTAL: <ST.SALARY "
" "
ON TABLE SET ACCESSHTML 508
ON TABLE SET STYLE *
TYPE=REPORT, GRID=OFF, FONT=ARIAL, SIZE=8,$
TYPE=HEADING, LINE=1, STYLE=BOLD, $
TYPE=HEADING, LINE=2, STYLE=ITALIC, $
ENDSTYLE
END
```

Centering a Heading

The report is:

PAGE 1

SALARY REPORT
FOR COMPANY EMPLOYEES

<u>DEPARTMENT</u>	<u>SALARY</u>	<u>EMP ID</u>
MIS	\$8,650.00	543729165
	\$9,000.00	543729165
	\$13,200.00	112847612
	\$17,750.00	117593129
	\$18,480.00	117593129
		219984371
	\$21,780.00	326179357
	\$25,775.00	818692173
	\$27,062.00	818692173
MIS SALARY TOTAL:	\$160,177.00	
PRODUCTION	\$9,050.00	119265415
	\$9,500.00	119265415
	\$10,000.00	071382660
	\$11,000.00	071382660
	\$15,000.00	451123478
	\$16,100.00	451123478
	\$21,120.00	126724188
	\$24,420.00	123764317
	\$26,862.00	123764317
	\$29,700.00	119329144
PRODUCTION SALARY TOTAL:	\$172,752.00	

FOR INTERNAL USE ONLY

WebFOCUS adds the COLSPAN attribute to the <TD> tags for the heading rows in the heading table. There are three columns in the output, so COLSPAN=3:

```
<TR>  
<TD NOWRAP COLSPAN=3 ALIGN=CENTER>  
<STRONG>SALARY REPORT</STRONG></TD>  
</TR>
```

Summarizing Report or Graph Information

How to:

Summarize Report or Graph Information

Example:

Summarizing Report Information

Summarizing Graph Information

A WebFOCUS StyleSheet supports a SUMMARY attribute for reports and graphs. This attribute maps to the HTML <TABLE SUMMARY> tag in the HTML report output. A description of the report or graph is placed into a summary object inside the HTML table in the source code.

Including a description of a report or graph supports accessibility.

Syntax: **How to Summarize Report or Graph Information**

Add a SUMMARY declaration to the WebFOCUS StyleSheet:

```
SUMMARY='description'
```

where:

description

Is the description of the report or graph. Enclose the description in single quotation marks ('). The length can be no longer than 500 characters. SUMMARY is also supported for reports in which SET ACCESSHTML is OFF.

Note: SUMMARY is not supported with SET GRAPHEDIT=OFF. It only works with server side graphs (SET GRAPHEDIT=SERVER, the default configuration for the WebFOCUS Reporting Server).

Example: Summarizing Report Information

The SUMMARY attribute that describes report information appears in boldface:

```
TABLE FILE GGSALES
SUM UNITS DOLLARS
BY CATEGORY BY PRODUCT
BY REGION
WHERE CATEGORY EQ 'Food';
WHERE REGION EQ 'Midwest' OR 'Northeast';
ON TABLE SET PAGE-NUM OFF
ON TABLE SET ACCESSHTML 508
ON TABLE SET STYLE *
SUMMARY='This report lists food product sales in the Midwest and
Northeast', $
TYPE=REPORT, GRID=OFF, $
ENDSTYLE
END
```

The text assigned in the SUMMARY StyleSheet attribute is placed in the HTML SUMMARY attribute for the HTML <TABLE> tag.

```
<TABLE CELLPADDING=1
SUMMARY="This report lists food product sales in the Midwest and
Northeast">
```

The report is:

<u>Category</u>	<u>Product</u>	<u>Region</u>	<u>Unit Sales</u>	<u>Dollar Sales</u>
Food	Biscotti	Midwest	86105	1091727
		Northeast	145242	1802005
	Croissant	Midwest	139182	1751124
		Northeast	137394	1670818
	Scone	Midwest	116127	1495420
		Northeast	70732	907171

Example: Summarizing Graph Information

The SUMMARY attribute that describes graph information appears in boldface:


```
GRAPH FILE GGSALES
SUM UNITS DOLLARS
BY PRODUCT
ON GRAPH SET ACCESSHTML 508
ON GRAPH SET STYLE *
SUMMARY='This graph shows unit and dollar sales for all products', $
ENDSTYLE
END
```

The text assigned in the SUMMARY StyleSheet attribute is placed in the ALT attribute on the HTML tag.

The HTML source code is:

```
<IMG SRC="/ibi_apps/WFServlet?&PG_Func=GETBINARY&PG_File=wbqqfpxc.png"
USEMAP="#vtdlnpj0" BORDER=0 ALT="This graph shows unit and dollar sales
for all products">
```

The graph is:



WebFOCUS Graphs are images with an HTML wrapper. In order to meet accessibility standards, there must be an ALT tag with information on what is being displayed on the chart/graph. If a report has a chart and the report needs to be 508 compliant, there should also be a tabular report with the data being summarized by the chart.

Describing Drill-Down Information

How to:

Describe Drill-Down Information

Example:

Describing Drill-Down Information

You can provide a description of a drill-down report using the ALT attribute in a WebFOCUS StyleSheet. In the HTML report output, the description is placed in the HTML TITLE attribute on the <HREF> tag for the drill-down hyperlink.

Including a description of the detail component of a drill-down report supports accessibility.

Syntax: How to Describe Drill-Down Information

```
ALT='description'
```

where:

description

Is the description of the drill-down information in a report. Enclose the description in single quotation marks ('). The length can be no longer than 256 characters. ALT is also supported for reports in which SET ACCESSHTML is OFF.

Example: Describing Drill-Down Information

The ALT attribute that maps to the drill-down hyperlink of a report appears in boldface:

```
TABLE FILE GGSALES
SUM UNITS DOLLARS
BY CATEGORY
HEADING
"* Click category to see product and regional breakdowns."
ON TABLE SET PAGE-NUM OFF
ON TABLE SET ACCESSHTML 508
ON TABLE SET STYLE *
TYPE=REPORT, GRID=OFF, $
TYPE=DATA, COLUMN=CATEGORY, FOCEXEC=REGSALES (TYPE=CATEGORY),
ALT='This will run a report that shows product sales by region', $
ENDSTYLE
FOOTING
"This report was created on &DATE ."
END
```

The HTML source code is:

```
<TD NOWRAP>
<A
HREF="/ibi_apps/WFServlet?IBIF_webapp=/ibi_apps&IBIC_server=EDASERVE&IBIW
F_msgviewer=OFF&IBIAPP_app=doc769%20doc769%20baseapp%20ibidemo%20ibisamp%
20baseapp&IBIF_ex=REGSALES&CLICKED_ON=&TYPE=Coffee" TITLE="This will run
a report that shows product sales by region">Coffee</A>
</TD>
```

The report is:

* Click category to see product and regional breakdowns.

<u>Category</u>	<u>Unit Sales</u>	<u>Dollar Sales</u>
Coffee	1376266	17231455
Food	1284845	17229333
Gifts	927880	11695502

This will run a report that shows product sales by region

This report was created on 05/12/09 .

Adding Descriptive Text to an Image

How to:

Add Descriptive Text to an Image

Example:

Adding Descriptive Text to an Image

The ALT attribute in a WebFOCUS StyleSheet adds descriptive text to an embedded image in an HTML report. The ALT attribute in the StyleSheet generates an HTML ALT attribute on the tag.

Including a description of an image supports accessibility.

Microsoft® Internet Explorer® shows the descriptive text in a box when you hold the mouse cursor over the image.

By default, the browser displays the descriptive text if the image loader is turned off or if the browser does not display images.

Syntax: How to Add Descriptive Text to an Image

`ALT='description'`

where:

`description`

Is a brief description of the image. Enclose the description in single quotation marks ('). The length can be no longer than 256 characters. ALT is also supported for reports in which SET ACCESSHTML is OFF.

Example: Adding Descriptive Text to an Image

This request adds the Gotham Grinds logo to a report heading. It uses the WebFOCUS StyleSheet ALT attribute to add descriptive text (Gotham Grinds Company Logo) that identifies the image.

```
TABLE FILE GGORDER
ON TABLE SUBHEAD
"PRODUCTS ORDERED ON 08/01/96"
SUM QUANTITY AS 'Ordered Units' BY PRODUCT
WHERE PRODUCT EQ 'Coffee Grinder' OR 'Coffee Pot';
WHERE ORDER_DATE EQ '08/01/96';
ON TABLE SET ACCESSHTML 508
ON TABLE SET PAGE-NUM OFF
ON TABLE SET STYLE *
TYPE=TABHEADING, IMAGE=/IBI_HTML/gotham.gif, IMAGEBREAK=ON,
ALT='Gotham Grinds Company Logo', $
ENDSTYLE
END
```


When you run the request and hold the mouse cursor over the image in Microsoft Internet Explorer, the descriptive text displays in a box, similar to the following:



PRODUCTS ORDERED ON 08/01/96

Product	Ordered Units
Coffee Grinder	2493
Coffee Pot	3100

WebFOCUS generates the following HTML code for the image:

```
<IMG SRC="/ibi_html/gotham.gif"
ALT="Gotham Grinds Company Logo">
```

On Demand Paging (ODP) WebViewer Support

In this section:

- Generating Section 508 Compliant ODP Output
- Page Breaks for Section 508 Compliant WebViewer Output
- Limitation Using the SHOW ALL Button in the WebViewer

The On Demand Paging report viewer (WebViewer) is Section 508 compliant. It presents all reports (both standard ODP and Section 508 compliant reports) using page by page navigation as defined by the physical pages within the report. Page breaks are defined through a combination of the lines per page setting (SET LINES=) and the sort field breaks in the report request (ON *sortfield* PAGE-BREAK). When both options are used together, a new page break is generated when either the defined lines per page is reached or a new sort value is encountered.

Report requests for HTML Section 508 compliant reports that do not have the On Demand Paging (ODP) viewer enabled ignore sort field page breaks (ON *fieldname* PAGE-BREAK). Report requests that send the output to the WebViewer present page breaks in a compliant manner.

Generating Section 508 Compliant ODP Output

Reference:

Screen Reader Settings for Use With the WebViewer

To qualify as an ODP Section 508 compliant report, the procedure must contain commands to enable the WebViewer and to make the output Section 508 compliant.

Issue the following command to generate Section 508 compliant output either for all requests or for a single request:

```
SET ACCESSHTML = 508 or ON TABLE SET ACCESSHTML 508
```

For upward compatibility for releases prior to Version 7 Release 7.04, the following legacy commands are still available:

```
SET ACCESSIBLE = 508 or ON TABLE SET ACCESSIBLE 508
```

Issue the following command to send report output to the ODP WebViewer, either for all requests or for a single request:

```
SET WEBVIEWER = ON or ON TABLE SET WEBVIEWER ON
```

Reference: Screen Reader Settings for Use With the WebViewer

In order for screen readers like JAWS to work properly with the WebViewer, the following settings must be in effect. For more information about setting these options, refer to the documentation for JAWS or for the screen reader you use:

- Virtual PC Cursor. *Use Virtual PC Cursor* must be checked.
- Forms Mode Auto Off. *Forms Mode Auto Off* must be checked.
- Graphics Setting. *Use Alt* must be checked.
- Button Text Options. *Use Title* must be checked.

Page Breaks for Section 508 Compliant WebViewer Output

For output designated as accessible and ODP enabled, page breaks defined in the report are implemented and respected. The ODP WebViewer presents the report one page at a time using the defined page breaks as follows:

- Each time the defined LINE limit is reached, a physical page break is inserted.


- ❑ Each time a new value is encountered for the designated sort field, a physical page break is inserted. Note that page breaks generated by sort field breaks can create pages shorter than the designated LINE limit.
- ❑ Standard page break functionality including the insertion of page headings and footings are implemented at each physical page break.

No additional syntax is required to implement this feature.

Limitation Using the SHOW ALL Button in the WebViewer

Within the ODP WebViewer, the *SHOW ALL* button presents the report as a continuous document without ODP navigation controls. In Section 508 mode, the *SHOW ALL* functionality presents the report as a single continuous HTML page, repeating the headings and footings for each designated physical page. These repetitive elements make the report non-compliant.

When a Section 508 accessible report with sort field page breaks is opened in the ODP WebViewer, the ALT tag associated with the *SHOW ALL* button will inform the user that this functionality may make access through a reader difficult. This information will be available to all users in the tooltip presented when the mouse pointer is placed on the button and as part of the text read by the accessibility readers:



Accessibility Limitations

In this section:

- ACROSS and OVER Reports
- Defining a Scrollable Area in a Report: HFREEZE
- FORMAT HTMTABLE Report Output

This topic describes limitations for accessibility.

Reports using the following elements are not accessible:

- ❑ ACROSS and OVER reports.
- ❑ Reports that generate FORMAT HTMTABLE output.
- ❑ Reports using the HFREEZE StyleSheet attribute.

- ❑ Accordion reports.
- ❑ Financial Modeling Language (FML).
- ❑ Table of Contents.
- ❑ OLAP.
- ❑ Webviewer (On Demand Paging) *SHOW ALL* button.

In addition, text included or designed in HEADING, FOOTING, SUBHEAD, SUBFOOT, RECAP, SUMMARIZE, and RECOMPUTE commands do not generate row header tags.

ACROSS and OVER Reports

Reports that use the ACROSS sort phrase and OVER sort phrase are not accessible, so you must use the BY sort phrase.

For more information about the ACROSS, OVER, and BY phrases, see the *Creating Reports With WebFOCUS Language* manual.

Defining a Scrollable Area in a Report: HFREEZE

The HFREEZE attribute in a StyleSheet enables you to freeze headings and/or footings, creating a scrollable area in the body of the report. This attribute does not create accessible report output and is ignored in a report that sets the ACCESSHTML parameter to 508.

FORMAT HTMTABLE Report Output

In requests using FORMAT HTMTABLE, the headings are not aligned properly with the body. WebFOCUS HTML Section 508 accessible reports include JavaScript that does the width calculation and alignment processing and ties it to the <BODY> tag of the HTML report output. FORMAT HTMTABLE produces only a table, not a complete HTML form, so there is no <BODY> tag.